Calcolo del valore di π con il metodo di Archimede

Il progetto seguente illustra la definizione del numero irrazionale π come elemento separatore di due successioni di numeri razionali, una crescente e l'altra decrescente.
Gli elementi della prima successione sono i valori ottenuti dai rapporti tra i perimetri dei poligoni inscritti in una circonferenza e la lunghezza del suo diametro, all'aumentare del numero di lati dei poligoni stessi. Quelli della seconda successione sono invece i valori ottenuti dai rapporti tra i perimetri dei poligoni circoscritti alla stessa circonferenza e la lunghezza del suo diametro, all'aumentare del numero di lati dei poligoni.
Il metodo di Archimede è un metodo iterativo, mediante il quale il lato di un poligono inscritto o circoscritto alla circonferenza, è calcolato utilizzando la lunghezza del lato del poligono avente un numero di lati inferiore e determinato nel passaggio precedente.
Consideriamo per esempio una circonferenza di raggio unitario; indicato con n il numero di lati di un poligono regolare in essa inscritto, e con l_n la lunghezza del lato di questo poligono, si ha che la lunghezza l_{2n} del lato del poligono regolare inscritto nella stessa circonferenza, avente un numero doppio di lati, è data dalla formula: $l_{2n} = \sqrt{2 - \sqrt{4 - l_n^2}}$

Considerando la stessa circonferenza di raggio unitario, e indicato con L_n la lunghezza del lato di un poligono regolare circoscritto, la lunghezza L_{2n} del lato del poligono regolare, circoscritto alla stessa circonferenza, con un numero doppio di lati, si calcola mediante la formula:

$$L_{2n} = \frac{\sqrt{16 + 4 \cdot L_n^2} - 4}{L_n}$$

Il primo elemento delle due successioni dipende dal numero di lati del poligono di partenza:
- se per esempio il primo poligono è un esagono regolare, il lato dell'esagono inscritto è uguale al raggio, quindi vale 1, mentre il lato dell'esagono circoscritto è pari a $\frac{2}{\sqrt{3}}$;
- se si considera come primo poligono un quadrato, il lato del quadrato inscritto è dato da $\sqrt{2}$, mentre il lato del quadrato circoscritto è pari a 2.

Il perimetro di ciascun poligono è dato dal prodotto del numero di lati per la lunghezza del lato; ciascun elemento delle due successioni si ottiene dal rapporto tra il perimetro e la lunghezza del diametro della circonferenza, che vale 2 in quanto il raggio è unitario.

Preparazione del foglio di lavoro

Aperto un nuovo foglio di lavoro di Excel, dalla cella A1 inseriamo una casella di testo con il titolo: "*CALCOLO DI π*".
Nella cella A6 scriviamo in grassetto: "*Raggio della circonferenza = 1*".
Nella cella A9 scriviamo al centro: "*Numero dei lati*".
Nella cella B9 scriviamo al centro: "*Lato del poligono inscritto*".
Nella cella C9 scriviamo al centro: "*Perimetro del poligono inscritto*".
Nella cella D9 scriviamo al centro: "*Lato del poligono circoscritto*".
Nella cella E9 scriviamo al centro: "*Perimetro del poligono circoscritto*".
Nella cella F9 scriviamo al centro: "*Perim. inscritto/diametro*".
Nella cella G9 scriviamo al centro: "*Perim. circoscritto/diametro*".
Nella cella F7 scriviamo al centro: "*Successione crescente*".
Nella cella G7 scriviamo al centro: "*Successione decrescente*".

Selezioniamo la riga 9 e assegniamo alle celle l'allineamento **Al centro** sia verticale che orizzontale, con **Testo a capo**.
Uniamo le celle F7 ed F8 e assegniamo loro l'allineamento **Al centro** sia orizzontale che verticale; in modo analogo diamo lo stesso formato alle celle G7 e G8.

Inseriamo le formule

Nella cella A10 scriviamo "6": consideriamo come primo poligono l'esagono.
Scriviamo le formule nelle rispettive celle:
in A11: = 2*A10; dopo aver selezionato la cella A11, trasciniamo il quadratino di riempimento fino alla cella A40, per ottenere il numero dei lati di ciascun poligono, ciascuno il doppio del precedente;
in B10: 1;
in B11: = RADQ(2-RADQ(4-B10^2)); selezionata la cella B11, trasciniamo il quadratino di riempimento fino alla cella B40 per copiare la formula della cella B11;
in C10: = A10*B10; selezionata la cella C10, trasciniamo il quadratino di riempimento fino alla cella C40;
in D10: = 2/RADQ(3);
in D11: = (RADQ(16+4*D10^2)-4)/D10; selezionata la cella D11, trasciniamo il quadratino di riempimento fino alla cella D40;
in E10: = A10*D10;
in F10: = C10/2;
in G10: = E10/2;
selezionate le celle E10, F10, G10, trasciniamo il quadratino di riempimento fino alla riga 40.
Utilizziamo il **Bordo casella spesso** per distinguere le varie parti delle due successioni.

Nelle colonne F e G otteniamo le due successioni dei rapporti tra i perimetri e il diametro. Notiamo che le due successioni si mantengono separate, la prima crescente e la seconda decrescente, fino alla riga 22, corrispondente a un poligono di 24.576 lati, per poi intersecarsi e non essere più monotone. Ciò è dovuto al fatto che l'errore commesso nei calcoli e causato dall'insieme limitato di numeri non reali a disposizione di Excel, è maggiore della lunghezza dei lati dei poligoni inscritti e circoscritti alla circonferenza.
Possiamo concludere che il valore verso cui tendono le due successioni fino a 24.576 lati, è quello del numero irrazionale π.

	A	B	C	D	E	F	G
1–4	**CALCOLO DI π**						
5							
6	Raggio della circonferenza = 1						
7–8						Successione crescente	Successione decrescente
9	Numero dei lati	Lato del poligono inscritto	Perimetro del poligono inscritto	Lato del poligono circoscritto	Perimetro del poligono circoscritto	Perim. inscritto/diametro	Perim. circoscritto/diametro
10	6	1,0000000000	6,0000000000	1,1547005384	6,9282032303	3,0000000000	3,464101615138
11	12	0,5176380902	6,2116570825	0,5358983849	6,4307806183	3,1058285412	3,215390309173
12	24	0,2610523844	6,2652572266	0,2633049952	6,3193198842	3,1326286133	3,159659942098
13	48	0,1308062585	6,2787004061	0,1310869256	6,2921724303	3,1393502030	3,146086215131
14	96	0,0654381656	6,2820639018	0,0654732208	6,2854291993	3,1410319509	3,142714599646
15	192	0,0327234633	6,2829049446	0,0327278443	6,2837461000	3,1414524723	3,141873049980
16	384	0,0163622792	6,2831152158	0,0163628268	6,2833254941	3,1415576079	3,141662747055
17	768	0,0081812081	6,2831677843	0,0081812765	6,2832203532	3,1415838921	3,141610176600
18	1.536	0,0040906126	6,2831809265	0,0040906211	6,2831940686	3,1415904632	3,141597034323
19	3.072	0,0020453074	6,2831842121	0,0020453084	6,2831874976	3,1415921060	3,141593748817
20	6.144	0,0010226538	6,2831850332	0,0010226539	6,2831858557	3,1415925166	3,141592927874
21	12.288	0,0005113269	6,2831852373	0,0005113269	6,2831854512	3,1415926186	3,141592725623
22	24.576	0,0002556635	6,2831852906	0,0002556635	6,2831853435	3,1415926453	3,141592671742
23	49.152	0,0001278317	6,2831852906	0,0001278317	6,2831852378	3,1415926453	3,141592618901
24	98.304	0,0000639159	6,2831852906	0,0000639159	6,2831853435	3,1415926453	3,141592671742
25	196.608	0,0000319579	6,2831852906	0,0000319579	6,2831838718	3,1415926453	3,141591935882

Inseriamo ora un **Grafico a linee con indicatori**, ponendo sull'asse orizzontale le celle da A10 ad A22 e sull'asse verticale le due successioni, da F10 a F22, e da G10 a G22. In questo grafico è evidente che le due successioni di numeri sono separate e convergenti.



Rinominiamo il Foglio1 con il nome "*Da esagono*".

Calcoliamo ora le due successioni partendo dal quadrato.
Copiamo il foglio *Da esagono* prima del Foglio2; scriviamo nelle celle:
in A10: il numero "4";
in B10: =RADQ(2); è la formula per calcolare la lunghezza del lato del quadrato inscritto nella circonferenza di lato unitario;
in D10: il numero "2"; è la lunghezza del lato del quadrato circoscritto.

Nelle colonne F e G si ottengono così le due successioni che convergono fino alla riga 24, corrispondente a poligoni di 65.536 lati.
Possiamo disegnare anche per queste due successioni il corrispondete grafico; rinominiamo il foglio con il nome "*Da quadrato*" e salviamo il lavoro.